ANTHROPOLOGIE ✶ XXVIII/1 ✶ 1990

JOSEF KOPŘIVA, KAREL ČECHOVSKÝ

DETERMINATION OF HEATH-CARTER SOMATOTYPE AND SOMATOTYPE DISPERSION INDEX USING A COMPUTER

ABSTRACT — Heath-Carter somatotype is widely used in sport anthropology and kinesiology. The conversion of anthropometric data into somatotype components by means of tables is time-consuming and inaccurate. It is possible to substitute the tables by simple approximating functions and to compute the components of the somatotype with computer. If several somatotypes are known, it is sometimes useful to compute the somatotype dispersion index. The programs in BASIC for this purpose are given.

KEY WORDS: Somatotype — Heath-Carter somatotype — Computer and somatotyping — Index of somatotype dispersion.

INTRODUCTION

In 1967 introduced Heath and Carter anthropometric somatotype has been to describe body type in man. At present it is frequently used typological method in cross-sectional studies. It expresses best the actual phenotype or morphologic structure of both men and women as well as of extreme types of body due to its open-ended scale. These advantages outweigh some limitations, for instance the neglect of correlation between somatotype components and age. It must be considered also that the somatotype components express the present status of phenotype and they often do not correspond to primary patterns of genotype.

In practice the main difficulty is in laborious and time-consuming determination of somatotype components, especially of the mesomorphic component, by means of tables.

METHOD

Standard methods for measuring the skinfolds, bone and muscle measures, height and weight are described by Heath-Carter (1967).

The conversion of these data into somatotype components is made by means of special tables (Hebbelinck, Ross, 1974). It is possible to approximate these tables by simple functions, mostly linear. Only in case of the endomorphic componen, it is the combination of the linear and fractional functions, for which the parameters can be calculated by the least-square method.

I. The endomorphic component is a function of the sum of three skinfolds.
For $S < 31.3$ mm endomorphy = $= (0{,}125 \,.\, S) - 0.625$
For $S \geq 31.3$ mm endomorphy = $= 24.447 \,.\, S/(195.207 + S)$
The first function is exact ($r = 1$), the second approximating with $r = 0.9992$.

II. The mesomorphic component can be calculated from the serial number of lines in tables:

for body height — H	$n_1 = (H - 78.7)/3.81132$
for humerus (epicond. width)	$n_2 = (U - 2.87)/0.14566$
for femur (epicond. width)	$n_3 = (F - 4.09)/0.207736$
for circum. of biceps	$n_4 = (B - 13.1)/0.664151$

for circum. of calf $n_5 = (C - 15.3)/0.771358$.

Then mesomorphy = $[(n_2 + n_3 + n_4 + n_5 - 4 \cdot n_1)/8] + 4$.

This substitution of table is exact ($r = 1$).

III. The ectomorphic component is determined from the ponderal index

$$I = H/\sqrt[3]{M}$$

where: H — body height in cm, M — body weight in kg.

From this index, ectomorphy = $(0.70671 \cdot I) - 27.434$,

exactly ($r = 1$).

Remark: from the circumferences we must subtract the thickness of the corresponding skinfolds.

In calculations of the endomorphic component for ages between 8 and 14 we have introduced the height correction 170,18 cm (Ross and Day, 1972; Bláha et al., 1986). This is necessary because in this age the subcutaneous fat increments are great and the sum of three skinfolds may cause the estimation of the endomorphic component to be too law. The comparison of corrected and uncorrected values of endomorphy in several groups of athletes and non-athletes is in *Table 1.*

Even the statistically significant corrections did not change the dominancy of components. Therefore Hebbelinck and Ross (1974) do not consider this correction as necessary.

We used the same age limits 8 and 14 for boys and girls, in contrast to the study of Bláha et al. (1986). This can be justified by results of longitudinal study of M.Bouchalová (1987). For boys of 14 there comes the change of the correlation of skinfold thickness with body height from positive to negative, as well as the end of pre-pubescent growth of skinfolds.

Because of smooth approximation it is not necessary to correct the mesomorphic component for values not given in tables. The program for the determination of somatotype components in BASIC is in *Figure 1.*

TABLE 1. *The comparison of the endomorphic component with and without correction*

GROUP	n	AGE		ENDOM. COMP. (uncorrected)		ENDOM. COMP. (corrected)		t-test
		$\bar{x}$	s	$\bar{x}$	s	$\bar{x}$	s	
Modern gymnasts	12	11.5	1,3	0,8	0,3	1,0	0,2	1.748 N. S.
Primary school pupils (boys)	39	8,7	0,3	1,4	0,7	1,9	0,9	2,386*
Primary school pupils (girls)	22	8,7	0,5	2,0	0,9	2,6	1,0	2,048*
Tennis players (boys)	23	13,0	0,4	1,5	0,3	1,6	0,4	0,394 N. S.
Tennis players (girls)	12	12,9	0,5	1.8	0,4	2,0	0.6	0,550 N. S.
Ice-hockey players (boys)	29	8,5	0,3	1,4	0,9	1,8	1,0	1,631 N. S.

* P < 0,05
N. S. non-significant difference

TABLE 2. *The comparison of the Heath—Carter somatotypes determined from tables and calculated using a computer. Somatotype dispersion index*

GROUP	n	AGE		SOMATOTYPE (table)			SOMATOTYPE (calculated)			SDI
University students (males)	20	$\bar{x}$	19.9	2,4	5,0	2,9	2,4	5,1	3,1	4,13
		$s_{\bar{x}}$	0,3	0,2	0,3	0,3	0,2	0,3	0,3	
University students (females)	20	$\bar{x}$	19,0	4,2	4,3	2,3	4,2	4,4	2,5	3,27
		$s_{\bar{x}}$	0,2	0,2	0,2	0,2	0,2	0,2	0,2	
Gymnasts (males)	14	$\bar{x}$	18,5	0,9	6,6	2,8	0,8	6,4	3,0	2,85
		$s_{\bar{x}}$	0,8	0,1	0,2	0,3	0,1	0,3	0,3	
Gymnasts (females)	26	$\bar{x}$	16,4	1,6	4,7	3,2	1,6	4,7	3,2	2,68
		$s_{\bar{x}}$	0,3	0,2	0,1	0,2	0,2	0,1	0,2	

Fig.1

```
10 REM*HEATH-CARTER SOMATOTYPE*
15 INPUT"AGE [years]=";A
20 INPUT"HEIGHT [cm]=";H
30 INPUT"SUM OF SKINFOLDS [mm]=";S
35 IF A)=8 AND A(15 THEN S=S*(170.18/H)
40 IF S(31.3 THEN LET EN=0.125*S-0.625:GOTO 60
50 EN=24.447*S/(195.207+S)
60 PRINT:PRINT USING"ENDO=$.$$";EN:PRINT
65 PRINT"EPICONDYLAR WIDTHS [cm]"
70 INPUT"HUMERUS,FEMUR:";U,F
75 PRINT"GIRTHS CORRECTED FOR FAT [cm]"
80 INPUT"FLEXED ARM,CALF:";B,C
90 N1=(H-78.7)/3.81132
100 N2=(U-2.87)/0.14566
110 N3=(F-4.09)/0.207736
120 N4=(B-13.1)/0.664151
130 N5=(C-15.3)/0.771358
140 ME=(N2+N3+N4+N5-4*N1)/8+4
150 PRINT:PRINT USING"MESO=$.$$";ME:PRINT
160 INPUT"BODY MASS [kg]=";M
170 I=H/M^(1/3)
180 EC=0.709671*I-27.434
190 PRINT:PRINT USING"ECTO=$.$$";EC:PRINT
200 PRINT USING"PONDERAL INDEX=$$.$$";I:PRINT
210 PRINT"SOMATOTYPE"
220 PRINT USING"$.$";EN;:PRINT"-";
230 PRINT USING"$.$";ME;:PRINT"-";
240 PRINT USING"$.$";EC
250 END
**********************
RUN

AGE [years]=? 18
HEIGHT [cm]=? 172.5
SUM OF SKINFOLDS [mm]=? 17

ENDO=1.50

EPICONDYLAR WIDTHS [cm]
HUMERUS,FEMUR:? 7.2  9.3
GIRTHS CORRECTED FOR FAT [cm]
FLEXED ARM,CALF:? 28.9  37.2

MESO=5.07

BODY MASS [kg]=? 63.9

ECTO=3.19

PONDERAL INDEX=43.15

SOMATOTYPE
1.5-5.1-3.2
```

FIGURE 1. *The program for determination of Heath-Carter somatotype with computer.*

If somatotypes of several individuals are known, it is possible to calculate the Somatotype Dispersion Index — SDI (Ross and Wilson, 1973). The program for this index is in *Figure 2.* Remark: These programs were created using the Czechoslovak computer SAPI-1.

RESULTS AND DISCUSSION

We have tested the method with data of several groups of athletes and non-athletes of various age and sex groups. The results are in good agreement with the components obtained without a computer (*Table 2*).

In accord to Ross and Day's (1972) study on skiers we have found in groups of children aged 8—14 that the correction for height may increase the endomorphic component only about half a unit.

CONCLUSION

The determination of somatotype components or the somatotype dispersion index with a computer is simple and fast, the components are not quantified to the half units as with tables. The program is simple and short, and can be easily modified for various computers, programming languages or programmable calculators.

Fig.2

```
10 REM*Somatotype dispersion Index*
15 REM*(by Ross and Wilson)*
20 INPUT"NUMBER OF PERSONS=";P:PRINT
30 DIM N(P),M(P),K(P),X(P),Y(P)
40 FOR I=1 TO P
50 PRINT I;".EN,ME,EC:";
60 INPUT N(I),M(I),K(I)
70 NEXT I
75 PRINT
80 FOR I=1 TO P
90 A=A+N(I):B=B+M(I):C=C+K(I)
100 X(I)=K(I)-N(I)
110 Y(I)=2*M(I)-N(I)-K(I)
115 PRINT"X";I;"=";X(I),"Y";I;"=";Y(I)
120 NEXT I
130 A=A/P:B=B/P:C=C/P:PRINT
140 PRINT USING"MEAN SOMATOTYPE: $.$";A;:PRINT"-";
150 PRINT USING"$.$";B;:PRINT"-";
160 PRINT USING"$.$";C:PRINT
170 X=C-A:Y=2*B-C-A
180 PRINT USING"X=$.$";X;
190 PRINT USING"   Y=$.$";Y:PRINT
200 FOR I=1 TO P
210 D=D+SQR(3*(X(I)-X)^2+(Y(I)-Y)^2)
220 NEXT I
230 D=D/P
240 PRINT USING"SDI=$$.$$";D
250 END
**************************
RUN

NUMBER OF PERSONS=? 5

 1 .EN,ME,EC:? 3  5  2
 2 .EN,ME,EC:? 2  6  3
 3 .EN,ME,EC:? 2  5  4
 4 .EN,ME,EC:? 3  4  4
 5 .EN,ME,EC:? 3  5  4

X 1 =-1        Y 1 = 5
X 2 = 1        Y 2 = 7
X 3 = 2        Y 3 = 4
X 4 = 1        Y 4 = 1
X 5 = 1        Y 5 = 3

MEAN SOMATOTYPE: 2.6-5.0-3.4

X=0.8   Y=4.0

SDI= 2.49
```

FIGURE 2. *The program for calculation of Somatotype Dispersion index.*

REFERENCES

BLÁHA P. et al., 1986: *Anthropometric studies of the Czechoslovak population from 6 to 55 years. Czechoslovak spartakiade 1985 I/1.* ÚŠ ČSS 1985, Praha. 288 pp.

BOUCHALOVÁ M., 1987: *Vývoj během dětství a jeho ovlivnění.* Avicenum, Praha. 383 pp.

HEATH B. H., CARTER J. E. L., 1967: A modified somatotype Method. *Amer. J. Phys. Anthrop.*, 27: 57—74.

HEBBELINCK M., ROSS W. D., 1974: Body type and Performance. In: *Fitness, Health, and Work Capacity.* Ed. Leonard A. Larson. Pp. 266—283. Macmillan Publishing Co., Inc. New York.

ROSS W. D., DAY J. A. P., 1972: Physique and Performance in young Skiers. *J. Sport Med. and Phys. Fitness* 21: 30—37.

ROSS W. D., WILSON B. D., 1973: A Somatotype Dispersion Index. *Res. Quart.*, 44: 372—374.

Ing. Josef Kopřiva
Institute of Scientific
Instruments of ČSAV
Královopolská 147
612 64 Brno
Czechoslovakia

Dr. Karel Čechovský
Sports Medicine Clinic
Faculty Hospital
Pekařská 53
656 91 Brno
Czechoslovakia